
Topic 7. Double block designs: Latin Squares [ST&D 9.10-9.15] 
 
7. 1. Introduction 

The Randomized Complete Block Design is commonly used to improve the ability of an 
experiment to detect real treatment differences by partitioning a known source of 
variation (blocks) from the experimental error.  When this idea is extended to remove two 
different known sources of variations (i.e. two-way blocking), one resulting design is the 
Latin square.  In this design, the randomization of treatments is further restricted, 
relative to an RCBD, due to orthogonality requirements with both blocking variables.  In 
this discussion, we will refer to the two blocking variables in a general Latin Square as 
rows and columns.  The experiment is organized such that each treatment appears exactly 
once within each row and within each column.  Similar to RCBD's with one replication 
per block-treatment combination, Latin Squares have one replication per row-column-
treatment combination.  Unlike RCBD's, Latin Squares are not complete designs; they are 
an example of an incomplete blocked design. 

A Latin Square with treatments assigned to the first row and the first column in an 
alphabetical or numerical sequence is called a standard square.  Figure 1 shows the 
standard squares for 2 x 2, 3 x 3 and 4 x 4 designs.  Treatment levels are indicated by A, 
B, C, etc. 

2x2    4x4         

A B   A B C D  A B C D 
B A   B A D C  B C D A 

    C D B A  C D A B 

    D C A B  D A B C 

3x3             

A B C  A B C D  A B C D 

B C A  B D A C  B A D C 

C A B  C A D B  C D A B 

    D C B A  D C B A 
 

Figure 1.  Standard Latin squares for 2 x 2, 3 x 3 and 4 x 4. 
 
As the size of the square increases, the number of possible standard squares increases 
rapidly.  For any given square size (t x t), the number of possible different Latin squares 
is: 

(# of standard squares) (t!) (t - 1)! 
 
where t is the number of treatments.  For example, for t = 4, the number of total possible 
unique squares is: (4) (4!) (3!) = 576. 
 
7. 2. Examples 

Sometimes, part of the native variability among experimental units within an experiment 
can be attributed to two distinct factors.  For example, natural fertility and soil texture 

may vary in independent, systematic ways within a field.  When comparing three 
cultivars, these two sources of variation can be separated from the random variation of 
experimental error by arranging the levels of fertility and soil texture into blocks, as 
shown in Table 1. 
 

Table 1  A 3 x 3 Latin Square with treatments (cultivars C1, C2 and 
C3) randomized according to two blocking variables:  level of fertility 
(low, medium, high = columns) and gradient of soil texture (1, 2, 3 = 
rows). 

 
Soil Texture Fertility                                     

 L M H 
1 C3 C1 C2 
2 C2 C3 C1 
3 C1 C2 C3 

 
Note that Latin Squares require the same number of rows, columns, and treatments.  
Requiring the number of treatment levels to equal the number of levels is each of two 
separate blocking variables is a severe design constraint indeed.  Recognize, however, 
that this arrangement does not necessarily have to be a physical square.  For example, if 
trees in an orchard can be classified in terms of both size and distance from a windbreak, 
a Latin Square may be appropriate.  The diagram below is the field map of such an 
experiment, where the four treatment levels (A – D) were assigned to EU's according to a 
Latin Square: 
 
 
 
 
           D   A          B                C 
 
           C   B          D                A 
 
           C   D                     B                A 
 
           A   B          D                C                     
 
Figure 3.  Sixteen orchard trees classified by size and distance from a windbreak to which 
4 treatments (A, B, C and D) are assigned.  The size of a circle symbolizes a size of tree. 
 
As shown above, rows and columns do not necessarily refer to the spatial distribution of 
the experimental units. They can also refer to the order in which treatments are 
performed, to different pieces of equipment used in the experiment, to different 
technicians taking the measurements, etc. Latin squares are not exclusive of agricultural 
research:  

 Latin Squares are commonly used in sensory panels.  In these situations, the 
materials or treatments to be tested (e.g. different wines, fruits, etc.) can be 

Windbreak

blocked by evaluators (judges) and days so that differences among judges and 
differences among days of testing can be removed from experimental error. 

 Marketing studies sometimes use Latin Squares with days being rows and stores 
being columns. 

 Indeed, absolutely anything that can function as a block in an RCBD can function 
as a block in a Latin Square, assuming you can satisfy the strict requirements of 
the design (rows = columns = treatments; and each treatment appears exactly once 
in each row and exactly once in each column). 

 
Consider an experiment to test the effect of 4 different computer keyboard key spacings 
(S1 – S4) on typing speed.  Four people are recruited to use the four new keyboard 
designs.  Because different people type at different speeds, independent of keyboard 
design, the researchers decide to block by people.  For each test, the volunteer is provided 
a printed page of text which he or she must type.  Because the content of the text (e.g. 
word size, sentence complexity, etc.) may affect typing speed, independent of keyboard 
design, the researchers decide to use four different pages of text in the experiment and to 
block by text as well. 

 
 Volunteer 

Text 1 2 3 4 
1 S2 S3 S4 S1 
2 S3 S4 S1 S2 
3 S4 S1 S2 S3 
4 S1 S2 S3 S4 

 
 

7. 3. Randomization 

 Proper randomization is crucial to the validity of conclusions to be drawn from any 
experiment, Latin Squares included. Randomization is used both to neutralize the effects 
of any systematic biases as well as to meet the assumption of independence underlying 
the analysis. 

 

7. 3. 1. Manual method of randomization 

     The only restriction on the Latin square arrangement is that each treatment must 
appear in every row and every column of the table. The number of possible Latin squares 
increases rapidly as the size of the square increases, and it is necessary to choose one of 
these possible arrangements at random. A procedure that gives a satisfactorily 
randomized square is the following: 
 
1. Arbitrarily select a standard square for the number of treatments involved.  For 

example, if. 4 treatments are involved, the following standard square could be 
selected: 

    A B C D 
    B C D A 
    C D A B 
    D A B C 
 
2. From a table of random numbers or by some other procedure, select two sets of 

random numbers with size equal to the treatments involved. Then assign ranks to 
these sets of random numbers. For example: 

 

 Set 1 (for columns) Set 2 (for rows) 

Random number 9  1 6 4 8  5 3 7 
Rank 4  1 3 2 4  2 1 3 

   
3. Step 1: assign the ranks selected above to the rows and column headers of the 

standard square. Step 2: Order the rows according to the ranks. Step 3: Generate a 
new square according to the rankings of the columns. 

 
Step 1: assign ranks   Step 2: order rows   Step 3 order columns. 
 4 1 3 2   4 1 3 2   1 2 3 4 
4 A B C D  1 C D A B  1 D B A C 
2 B C D A  2 B C D A  2 C A D B 
1 C D A B  3 D A B C  3 A C B D 
3 D A B C  4 A B C D   4 B D C A 
 
 
4. Codes A, B, C, and D are then randomly assigned to the 4 treatments and the square 

shows the assignment of the treatments to the experimental units. 
 
7. 3. 2.  Automated method of randomization 

The following simple SAS program using PROC PLAN generates a template for a 5x5 
Latin square design. The row and column FACTORS are identified as ORDERED and 
the TREATMENTS as CYCLIC. The last two factor statements generate the randomized 
ranks to reorganize the rows and columns by hand as indicated in point 3 above.   
 
Proc Plan; 
 Factors rows = 5 ordered cols = 5 ordered; 
 Treatments tmts = 5 cyclic; 
 Factor r = 5; 
 Factor c = 5; 
Run;  
Quit; 

7. 4. The linear model 

The linear model for the Latin Square is, 

 yij(t)  =  + i  + j  +  (t)  +  ij      
 
where Yij represent the observation in the ith row and jth column, i  represents ith  row 
effect,  j represents j th column effect, and (t) represents the tth treatment effect. 
The parenthesis around the index “t” is due to the incomplete nature of this design; in any 
given row –column combination, only one treatment level occurs. A complete design 
would include all treatment combinations within each column – row combination (it will 
not be a Latin Square). 
 
In dot notation:  

 
 or,       
 
SStotal   =       SSrows         +     SScol.            +      SStrtmnt.        +   SSerror 
 
 

7. 5. ANOVA 

The ANOVA table for a Latin Square looks like this: 

Source df SS MS F 
Rows r - 1 SSR SSR/(r-1) MSR/MSE 
Columns r - 1 SSC SSC/(r-1) MSC/MSE 
Treatments r - 1 SST SST/(r-1) MST/MSE 
Error (r-1)(r-2) SSE SSE/(r-1)(r-2)  
Total r2-1 TSS   
 
The ANOVA from the SAS output for the example provided on ST&D page 230 is 
included below. This 4 x 4 Latin Square for yields of four wheat varieties shows highly 
significant differences among varieties, no significant differences among rows, and 
marginally significant differences among columns.  

Source          DF        SS         MS    F Value    Pr > F 
ROW              3      1.955     0.652      1.44     0.322 
COL              3      6.800     2.267      5.00     0.045 
TRTMT            3     78.925    26.308     58.03     0.000 
Error            6      2.720     0.453 
Total           15     90.400 

There are four identified sources of variation in this example, three due to the design 
(rows, columns and varieties, each with 3 degrees of freedoms) and one due to 
experimental error with (4-1)(4-2) = 6 degrees of freedom. For testing the hypotheses that 

there are no column, row, or treatment differences, the mean squares for each of these 
main effects are divided by the MSE. 

The following SAS program was used to produce the previous table. Note that the only 
change to the statements in Proc GLM compared to an RCBD is the inclusion of one 
extra classification variable:  

data STp230ls; 
  do row = 1 to 4; 
    do col = 1 to 4; 
      input trtmt $ yield @; 
      output; 
    end; 
  end; 
cards; 
C  10.5  D   7.7  B  12.0  A  13.2 
B  11.1  A  12.0  C  10.3  D   7.5 
D   5.8  C  12.2  A  11.2  B  13.7 
A  11.6  B  12.3  D   5.9  C  10.2 
 
proc GLM; 
  class row col trtmt; 
  model yield=row col trtmt; 
  means trtmt / lsd; 
  output out=resplot p=predyiel r=resiyiel; 

* Visual checking of the residuals; 
proc plot data=resplot; 
  plot resiyiel*predyiel=trtmt; 
 

* Normality of the residuals; 
proc univariate data=resplot normal; 
  var resiyiel; 
 

* Absence of multiplicative effects; 
* 3 separate Tukey non-additivity tests are required for the 3 possible 

interactions col*row, col*trt, and row*trt (only one shown below);   
proc GLM data= STp230ls; 
  class row trtmt; 
  model yield = row trtmt; 
  output out=tukrt p= prt   r= rrt; 
proc glm data=tukrt; 
       class row trtmt; 
       model yield = row trtmt prt*prt; 
 
* Homogeneity of variances; 
* This should be tested for the 3 one-way ANOVAS;   
proc GLM data= STp230ls; 
  class trtmt; 
  model yield = trtmt; 
  means trtmt / hovtest = levene; 
run; 
quit; 

7.6 Advantages and disadvantages of the Latin Square 

When two kinds of heterogeneity that are either due to the nature or arrangement of 
experimental units can be identified, the Latin Square is an effective design for 
partitioning those sources of variation from the experimental error.  Some disadvantages 
of the Latin Squares are: 

1. Severe design constrains: The number of treatments must be equal to the number 
of rows and columns. This restriction imposes a restriction for actual experimental 
work, particularly, for experiments with a large number of treatments. 

2. The design is only valid if there are no interactions among rows, columns, and 
treatments. When there is significant interaction between any two of these 
factors, the test is not very sensitive (the interactions will inflate the error term). 
To check this assumption all three interactions need to be tested. 

3. Squares smaller than 4 x 4 generally have too few replications for a desirable 
level of precision. This is reflected in the few degrees of freedom available for 
experimental error in the ANOVA. 

The analysis of variance for the completely randomized design (CRD), the randomized 
complete block design (RCBD), and the Latin Square design (LS) are shown 
schematically below.  Assume 5 treatments are assigned to 25 experimental units, with 
equal replication per treatment. 
 
       Total  Sum of Squares 

        df = 24 

CRD   SS Treatment   SS Experimental Error 
     df = 4       df = 20 

RCBD  SS treatment       SS block       SS Experimental error 
   df = 4         df = 4       df =16 

LS  SS treatment          SS Column     SS Rows             SS Experimental Error 
  df = 4      df = 4        df = 4       df = 12   

This diagram illustrates the loss of degrees of freedom from the experimental error as the 
chosen design becomes more complex.  Unless there is enough variation in the 
experimental error of the CRD that can be allocated to the chosen blocking variable(s) to 
compensate for the loss of degrees of freedom from the experimental error, the resulting 
design will be less efficient. 

A numerical example of the effect of the missing dfe in the critical F value: 

3x3 Latin Square      5x5Latin Square 
LS dfe=2   (RCBD dfe=4)   LS dfe=12 (RCBD dfe=16) 
 

LS F df 2, 2, 0.05=      19.0   LS F df 2, 12, 0.05=   3.89 

RCBD F df 2, 4, 0.05=  6.94   RCBD F df 2, 16, 0.05=   3.63 

The loss of dfe has a huge effect on the sensitivity of small LS! 

7. 7. Relative Efficiency 

 Section 9.3 of ST&D (page 237) provides a good discussion for the relative 
efficiencies of Latin Squares compared to RCBD. Once an experiment is conducted as a 
Latin Square, the researcher can ask the question of how efficient it was relative to an 
alternative RCBD for that particular set of data.  A Latin Square can be converted into an 
RCBD (i.e. simplified) in one of two ways, either by removing the rows as a blocking 
variable or by removing the columns as a blocking variable.  In either case, the estimated 
MSE for the alternative RCBD needs to be calculated.  In the example from ST&D page 
230, MSELS = 0.45, MSRLS = 0.65, and MSCLS= 2.27. 

If columns are removed as a blocking variable, the estimated MSE for the alternative 
RCBD will be a weighted average of the LS MSE and MSC: 

   
and the efficiency of the LS relative to the RCBD is: 

 
This indicates that the column grouping increased the precision of the model by an 
estimated 89%. This parallels the significant effect of columns identified in the F tests 
(see previous ANOVA table) 
 

If rows are removed as blocking variables, the estimated MSE for the alternative RCBD 
will be a weighted average of the LS MSE and MSR: 

 
and the efficiency of the LS relative to the RCBD is 

 
This indicates that the row blocking increased the precision of the model only by an 
estimated 4%. This parallels the non-significant effects of rows identified in the F tests 
(see previous ANOVA table). 

7. 7. Repeated Latin squares 

In cases where the number of treatments is small but considerable variability is expected 
to affect the experiment from two sources (two blocking variables), two or more Latin 
squares may be used to increase the degrees of freedom for experimental error.   
This strategy is known as repeated Latin Squares, where each individual Latin Square can 
be thought of as an independent replication of the entire experiment. 

 Repeated Latin Squares can relate to one another in four possible ways, depending on 
how the levels of the blocking variables relate across squares (i.e. experiments): 

1. They share the same rows and the same columns (common rows and cols) 
2. They share the same columns but not the same rows (common cols, independent 

rows) 
3. They share the same rows but not the same columns (common rows, independent 

cols) 
 
4. They share neither the same rows nor the same columns (independent rows and 

cols) 
 

Example: 
Three gasoline additives (TREATMENTS, A B & C) were tested for gas efficiency by 
three drivers (ROWS) using three different tractors (COLUMNS). The variable measured 
was the yield of carbon monoxide in a trap. The experiment was repeated twice. These 
replications can be: 
 

Case 1: same tractors (col) and drivers (col)  
PROC GLM; 
   class square col row treat; 
   model yield= square col row treat; 
 

Case 2: Using different drivers (row) but the same tractors (col) . 
PROC GLM; 
   class square col row treat; 
   model yield= square col row(square) treat; 

 

Case 3: Using different tractors (col) but the same drivers (row) 
PROC GLM; 
   class square col row treat; 
   model yield= square col(square) row treat; 

 

Case 4: Using different drivers (row) and tractors (col).  

             Columns and rows are within each square 

D1 
D2 
D3 

D4 
D5 
D6 

T1 T2 T3 

S2 

S1 

D1 
D2 
D3 

T1 T2 T3 T4 T5 T6

S2 S1 

D1 
D2 
D3 

T4 T5 T6

D4 
D5 
D6 

S2 

S1 

D1
D2
D3

T1 T2 T3 S1 
S2 

PROC GLM; 
  class square col row treat; 
  model yield= square col(square) row(square) treat; 
 
The complete SAS programs for each of these different cases will be covered in lab.  In 
the meantime, to convince yourself that SAS is doing the correct calculations for each of 
the above scenarios, take the following dataset: 
 

Square 1  Square 2 
B 26.0     C 25.0  A 21.3  C 32.4 B 28.7 A 25.8 
C 28.7 A 23.6 B 28.5  B 31.7 A 24.3 C 30.5 
A 25.3   B 28.4 C 30.1  A 24.9 C 29.3  B 29.2 

 
Run the previous example using row(square). Keep your output.  

Next, replace the IDs of the rows (1, 2, 3) in the second square by new IDs (4, 5, 6) to 
manually differentiate them from the rows in square one. In the Model statement change 
“row(square)” simply to “row”.   

You will get identical results in both analyses. The change of row numbers in the second 
square indicates SAS that there are 6 different tractors. The effect of the row(square) 
statement is exactly the same. It tells SAS that the numbers of the rows are only 
meaningful within each square.   
 

7. 8.  More complicated Latin squares (See Chapter 8 of Box, Hunter & Hunter, 
Appendix 8 included) 

To partition more than two independent sources of variability from the experimental 
error, more complicated designs such as the Graeco-Latin square (GLS) or a hyper- 
Graeco-Latin square (HGLS) may be useful. A GLS is a r x r pattern that permits the 
study of r treatments simultaneously with three different blocking variables. If even more 
blocking variables are included the design is called HGLS. 

An example of a 4 x 4 Graeco-Latin square could be the test of 4 different oil additives 
(treatments, A, B, C, and D) grouped by 4 different drivers (I, II, III, and IV), 4 different 
cars (1, 2, 3, and 4) and 4 different days (, , , ).  
 

         Car      
     1   2   3   4 
 
           I  A   B   C   D 
               II  B   A   D   C 
      Driver   III  C   D   A   B  
                    IV D   C   B   A 

 
Note in this particular case that a quadruple-replicated 4x4 Latin Square could 
accomplish a similar objective (Squares = Days) but would require four times the 

resources (64 driving tests, 16 on each of four days, vs. 16 driving tests, 4 on each of four 
days).  On the other hand, replicated Latin Squares with shared cars and drivers would 
provide much more power (Replicated LS dfe = 51 vs. GLS dfe = 3).  Here again we see 
the large penalty of small squares on power. 
 


